Credit Valuation Adjustment with Mathematica 10

Credit Valuation Adjustment [**CVA**] is an important valuation measure in finance, which determines how much financial institution needs to adjust a derivative contract value to account for a potential loss of positive exposure in case the counterparty gets into financial distress or defaults. In other words, CVA is the expected loss of a derivative contract due to counterparty default. As such, CVA is closely linked to exposure (presented in previous demonstrations) and extends the calculation in the credit direction. Formulae-wise, CVA can be expressed as: $CVA(u,T) = (1-R)\int_u^T B(t,T)EE\{t,T)dPD(t,T)dt$ where R is the counterparty recovery rate, B is the risk-free discount factor,EE is the expected exposure and dPD is the conditional default probability.
CVA, as the formula shows, is driven by four factors, so the calculation is quite involved. **Mathematica 10** is perfectly suited for this task since it provides all necessary components to compute CVA just in few steps. We demonstrate the case using Monte Carlo numerical method.

Let's assume the following case: financial institution enters into a 5Y semi-annual IR swap with 2.5% fixed rate with a counterparty whose 5Y CDS rate trades at 1.25% on 40% recovery. It wants to establish how much to charge for the CVA.
If we assume constant recovery and known risk-free rates, the problem reduces to two points: (i) determining future swap exposure and (ii) finding default probability. Both are unknown at the contract start time, so they need to be estimated.
Assume the following SDE processes for the swap and the CDS rates:

1. Swap rate: Log-Normal process with drift $\mu = 2.2\%$ and volatility σ=20%
2. CDS rate: Mean-revering square-root diffusion with rate of mean reversion $\theta = 1.5\%$, mean CDS rate $\beta = 1.5\%$ and volatility $\varphi = 5\%$
3. To make things more complicated, we further assume that the two stochastic processes are correlated with parameter $\rho = 13\%$.

We first build the term structure of discount factors from the given zero rates:

```
zrates = {0.012, 0.015, 0.0167, 0.0176, 0.019, 0.0201, 0.0223, 0.0235};
times = {0.25, 0.5, 1, 1.5, 2, 3, 4, 5};
zrobj = TemporalData[zrates, {times}];

intrate = Interpolation[zrobj["Path"], Method → "Spline"];

DF[x_] := Exp[-intrate[x] * x];

Plot[DF[x], {x, 0.25, 5}]
```



Then define two correlated random processes:

```
cW[ρ_] := Refine[ItoProcess[{{0, 0}, IdentityMatrix[2]}, {{n1, n2}, {0, 0}},
    t, {{1, ρ}, {ρ, 1}}], -1 < ρ < 1];

combprc = ItoProcess[{ds[t] == μ s[t] dt + σ1 * s[t] dn1[t],
    dc[t] == θ (β - c[t]) dt + σ2 * √Abs[c[t]] dn2[t]
   }, {s[t], c[t]}, {{s, c}, {s0, c0}}, t, {n1, n2} ≈ cW[ρ]];
```

Run MC simulation and convert CDS into hazard rates using approximation $h(t) = CDS(t)/(1-R)$.

```
simproc =
  RandomFunction[
   combprc /. {μ → 0.022, σ1 → 0.2, θ → 0.005, β → 0.015, σ2 → 0.05,
     ρ → 0.13, s0 → 0.025, c0 → 0.0125}, {0, y, dt}, 1000,
   Method → "StochasticRungeKutta"] ;
```

Process samples looks as follows:





From the simulated values, we get (i) positive exposure and (ii) expected hazard rate profiles:

```
Exposure[T_, S_, t_, v_] := (T - t) * Max[v - S, 0];

expswap = TimeSeriesMapThread[Exposure[5, 0.025, #1, #2] &, swapproc];

eeswap = TimeSeriesThread[Mean, expswap, ResamplingMethod → Interpolation];
eehaz = TimeSeriesThread[Mean, hazrate, ResamplingMethod → Interpolation];
ListLinePlot[eeswap, PlotStyle → {Red, Thick},
 PlotLabel → "Expected swap exposure"]
ListLinePlot[eehaz, PlotStyle → {Blue, Thick},
 PlotLabel → "Expected shazard rate"]
```

and visualise the values:





We then integrate path-wise to obtain the time-varying (i) swap expected positive exposure and (ii) mean hazard rate:

```
epe[x_] :=
  Mean[TimeSeriesWindow[eeswap, {0, Max[dt, x]},
    ResamplingMethod → Interpolation]];

Plot[epe[x], {x, 0, 5}, PlotStyle → {Purple, Thick},
 PlotLabel → "Swap EPE"]
```



Mean hazard rate:

```
exphazrate[x_] :=
  Mean[TimeSeriesWindow[eehaz, {0, Max[dt, x]},
    ResamplingMethod → Interpolation]];

Plot[exphazrate[x], {x, 0, 5}, PlotStyle → {Orange, Thick},
 PlotLabel → "Mean hazard rate"]
```



We next define the survival function – this is the probability that the counterparty survives (i.e. does not default) before the end of the contract:

$$S(u) = Exp\left[-\int h(u)du\right]$$

where $h(u)$ is the time-varying hazard rate defined previously

```
survfunc[x_] := Exp[-exphazrate[x] * x];

Plot[survfunc[x], {x, 0, 5}, PlotStyle → Brown,
 PlotLabel → "Survival probability"]
```



Since the swap pays in discrete intervals, we calculate the conditional default probabilities from the survival function above: $dPD(u) = S(u-1) - S(u)$

```
TemporalData[Table[{i, survfunc[i]}, {i, 0, 5, 1/2}]];
MovingMap[First[#] - Last[#] &, %, {2}];
ListLinePlot[%, PlotStyle → Magenta]
```



The conditional default probabilities are decreasing function of time.

Having obtained all components for the CVA calculation, we can now define a simple formula that completes the task:

```
CVAS[n_, fr_, r_] :=
  Module[{f, g, s}, f = TemporalData[Table[{i, survfunc[i]}, {i, 0, n, fr}]];
   g = MovingMap[First[#] - Last[#] &, f, {2}];
   s = (1 - r) * Sum[DF[i] * epe[i] * g["PathFunction"][i], {i, fr, n, fr}];
   s];
```

In discrete intervals we approximate the integral with the product summation of (i) discount factors, (ii) EPE and (iii) conditional default probability.
For the 5Y S/A swap above we get the CVA = **4.16** bp.

```
CVAS[5, 1/2, 0.4] * 10^4

4.16104
```

We can visualise how the CVA charge will evolve over time:

```
Table[{i, CVAS[i, 1/4, 0.4] * 10^4}, {i, 1/4, 5, 1/4}];
ListLinePlot[%, PlotLabel → "CVA as function of time", PlotStyle → Red,
  PlotTheme → "Web"]
```



We can also define other comparative static quite easily – for example - sensitivity of the CVA charge to the process correlation.
If the correlation rises from 13% => 50%, the CVA charge will increase to **4.74** bp. or 14%

```
CVAS[5, 1/2, 0.4] * 10^4

4.74152
```

This is in line with expectation – when both exposure and default rate raise, the loss will be naturally larger.